﻿import flash.external.*

//注册小铲子图片的click事件的响应函数，调用游戏中td_seedbank模块的select_del函数
function select_del()
{
	ExternalInterface.call("td_seedbank.select_del");
}
btnDel.addEventListener("click", this, "select_del");

//可种植植物卡片槽上面的卡片控件目前一共有6个
Card = [Card_0, Card_1, Card_2, Card_3, Card_4, Card_5]

//可种植植物卡片槽上面的卡片的click事件的响应函数，调用游戏中td_seedbank模块的select_seed函数，参数为卡片的下标
function select(e:Object)
{
	ExternalInterface.call("td_seedbank.select_seed", e.target.iJob);
}
//遍历当前存在的可种植植物卡片槽的卡片控件，设置下标并注册响应click事件的函数
for (i = 0; i < 6; i++ )
{
	Card[i].iJob = i;
	Card[i].addEventListener("click", this, "select");
}
//遍历当前存在的可种植植物卡片槽的卡片控件，生成限制显示区域的Mask
for  ( i = 0; i < 6; i++ )
{
	var Mask:MovieClip = Card[i].createEmptyMovieClip("Mask", Card[i].getNextHighestDepth());

	with (Mask)
	{
		lineStyle(0, 0); 
		beginFill(0, 10); 
		moveTo( 0, 0 ); 
		lineTo( 0, 84 ); 
		lineTo( 60, 84 ); 
		lineTo( 60, 0 ); 
		lineTo( 0, 0 ); 
		endFill();
	}
 	Card[i].borderCD.setMask(Mask);
}
/*设置显示卡片冷却的函数：具体的逻辑是，在swf生成的时候会给每个植物卡片附加一个Mask，作用是限定这个植物卡片控件中的所有的子控件的显示都不能超过这个Mask的范围，超过范围的会被截掉；
然后当冷却刚开始时冷却遮照图片的位置和植物图案的位置是重合的，但是随着冷却时间会慢慢地把冷却遮照图往上方（Y轴的负方向）移动，由于超出显示范围的冷却遮照图被Mask截掉了，
所以在画面表现上就像是冷却遮照渐渐地从下方消失不见*/
function set_cooldown(_index:Number, left:Number, total:Number)
{
	var cd_y:Number = int(84 * (total - left) / total);
	Card[_index].borderCD._y = -cd_y;
}

//可选择植物卡片箱子上的卡片控件，目前一共有14个
sCard = [choose.sCard_0, choose.sCard_1, choose.sCard_2, choose.sCard_3, choose.sCard_4, choose.sCard_5, choose.sCard_6, choose.sCard_7, choose.sCard_8, choose.sCard_9, choose.sCard_10, choose.sCard_11, choose.sCard_12, choose.sCard_13]

//可选择植物卡片箱子上的卡片控件的click事件的响应函数，调用游戏中td_seedbank模块的choose_card函数
function choose_card(e:Object)
{
	ExternalInterface.call("td_seedbank.choose_card", e.target.iJob);
}
//遍历当前存在的可选择植物卡片箱子上的卡片控件，设置下标并注册click事件的响应函数
for (i = 0; i< 10; i++)
{
	sCard[i].iJob = i;
	sCard[i].addEventListener("click", this, "choose_card");
}

//注册可选择植物卡片箱子上的开始游戏按钮的click事件的响应函数，调用游戏中td_seedbank模块的game_start函数
function game_start()
{
	ExternalInterface.call("td_seedbank.game_start");
}
choose.btnStart.addEventListener("click", this, "game_start");

function use_sun()
{
	ExternalInterface.call("td_seedbank.use_sun");
}
btnSun.addEventListener("click", this, "use_sun");